Kodaikanal Obs. Bull. Ser. A (1983) 3, 107-119

NUMERICAL EVALUATION OF SPHERICAL BESSEL AND RELATED FUNCTIONS

GHANSHYAM A. SHAH

ABSTRACT

Techniques for calculating the spherical Bessel functions of complex argument lying in any quadrant of the z-plane and arbitrary orders have been developed. The method applies to real, pure imaginary or complex arguments. Sample results are given for a range of arguments and for selected orders ranging from 0 to 200. It has been found that these functions for various quadrants are interrelated. We have noticed the following unusual result: provided that Im z is large compared to unity and Re z is not too near zero, the functions $J_{n+\frac{1}{2}}$ (z) and $Y_{n+\frac{1}{2}}$ (z) are interdependent in the sense that one can be derived from the other upto a certain maximum order n. Furthermore, if z lies on positive or negative y-axis, it has been found that the corresponding Bessel functions satisfy certain special properties.

Key words : Spherical Bessel functions, Special functions, numerical techniques.

Introduction

The Spherical Bessel and related functions are used frequently in many fields of physical sciences and engineering such as acoustic and electromagnetic scattering, plasma physics, fluid dynamics, nuclear physics, celestial mechanics, etc. The definition of such special functions are amenable to analytical treatment but in actual numerical evaluation one often encounters difficulties due to truncation errors and/or limitation on the range of available computer. In absorptive and dissipative systems, one is concerned with the Bessel functions of complex variable (or argument). Therefore, it is worthwhile to develop a general practical scheme for fast and accurate computations of these functions with arbitrary comlex argument lying in any quadrant of the complex plane and with arbitrary orders. In what follows, we present a method for evaluating the Riccati-Bessel functions and Bessel functions; the latter have been considered in the usual forms defined by Watson (1966) and McLachlan (1961) as well as by Abramowitz and Stegun (1964). A related important problem of electromagnetic scattering by a smooth, homogeneous and isotropic sphere, first solved by Mie (1908) involves the use of the spherical Bessel functions of the first and the second kinds of complex and real variables, respectively.

Analytical and Numerical Techniques

Let the complex variable (argument) be given by $z = x_1 + ix_2$. Here x_1 refers to x-coordinate and x_2 to the y-coordinate in the Argand diagram on the complex z-plane. The subscript n to a function indicates the order and a prime denotes the first derivative of the function with respect to the complex argument concerned. Following the notation of McLachlan (1961), z implies the independent variable or argument of a function, such as z in J_n (z). To avoid ambiguity, we define 'phase z' as $\theta = \tan^{-1} (x_2/x_1)$ in appropriate quadrant; this is the angle between the radial vector and positive real axis in the Argand diagram. The order n can take on positive integral values including zero.

Let us define Riccati-Bessel functions of the first kind, S_n (z), through the following recurrence relation :

$$S_n (z) = \left(\frac{2n-1}{z}\right) S_{n-1} (z) - S_{n-2}(z) \tag{1}$$

with the initial values

$$S_0 (z) = \sin z \tag{2}$$

and

$$S_1(z) = \frac{\sin z}{z} - \cos z \tag{3}$$

The Riccati-Bessel function of the second kind is defined through the recurrence relation :

$$C_n(z) = \left(\frac{2n-1}{z}\right) C_{n-1}(z) - C_{n-2}(z) \tag{4}$$

with the initial values

$$C_0(z) = \cos z \tag{5}$$

and

$$C_1(z) = \frac{\cos z}{z} + \sin z \tag{6}$$

The first derivatives of $S_n(z)$ and $C_n(z)$ are found from the recurrence relations :

$$S_n'(z) = -\frac{n}{z} S_n(z) + S_{n-1}(z) \tag{7}$$

$$C_n'(z) = -\frac{n}{z} C_n(z) + C_{n-1}(z) \tag{8}$$

The numerical evaluation of Riccati-Bessel functions remain stable if one performs downward recursion on $S_n(z)$ and upward recursion on $C_n(z)$ as revealed by earlier works by Shah (1967, 1977). Recent works by Gautschi (1967) and Temme (1976) conform to this conclusion. First we show the method for $S_n(z)$.

The computational work on the function $S_n(z)$ can be considerably simplified by using the following scheme based on the continued fractions. In other words, we adopt a method of downward recursion which assures the stability of the computations of the functions like $S_n(z)$, $J_{n+\frac{1}{2}}(z)$, $j_n(z)$ etc. Brouwer and Clemence (1961) have discussed an elegant method for evaluating higher order cylinder Bessel functions of the first kind of real variable. Here we have extended it to spherical Bessel and related functions of complex argument lying in any quadrant of the complex plane. The method has been applied successfully in developing the Fortran programs for Mie theories of scattering of electromagnetic wave by spheres (see, for example, van de Hulst 1957 ; Shah 1977) and by infinite single and composite cylinders (see Shah 1967, 1970).

Let us define a set of coefficients P_n and Q_n to represent the ratio of successive orders of $S_n(z)$ as follows:

$$\frac{S_n(z)}{S_{n-1}(z)} = P_n + i\,Q_n \tag{9}$$

From equation (1) one obtains :

$$\frac{S_n(z)}{S_{n-1}(z)} = \frac{2n-1}{z} - \frac{S_{n-2}(z)}{S_{n-1}(z)} \tag{10}$$

Substituting from equation (9), this reduces to :

$$\frac{1}{P_{n-1} + i\,Q_{n-1}} = \frac{2n-1}{z} - (P_n + i\,Q_n) \tag{11}$$

Let the square of absolute value of z be denoted by x, so that

$$x = |z|^2 = x_1^2 + x_2^2$$

We define the following auxiliary quantities for convenience :

$$\alpha_n = \frac{(2n-1)\,x_1}{x} - P_n \tag{12}$$

$$\beta_n = \frac{(2n-1)\,x_2}{x} + Q_n \tag{13}$$

A little algebraic manipulation among the equations (11-13) yields :

$$P_{n-1} = \frac{\alpha_n}{\alpha_n^2 + \beta_n^2} \tag{14}$$

$$Q_{n-1} = \frac{\beta_n}{\alpha_n^2 + \beta_n^2} \tag{15}$$

The equations (12-15) have been used to carry out downward recursions which provide the coefficients (P_n, Q_n) of all necessary orders for a given argument z. Initially, the starting value of N for n is assumed sufficiently large, preferably much larger than the largest z anticipated, so that it is permissible to set $P_N = 0$ and $Q_N = 0$, in the first instance. The sets of equations (12, 13) and (14, 15) are employed successively to calculate (P_{N-1}, Q_{N-1}), (P_{N-2}, Q_{N-2}), . . . (P_2, Q_2), (P_1, Q_1). The coefficients (P_n, Q_n) should be stored by treating them as dimensioned variables in the computer program. The initial guess of N may be varied and experimented upon to derive the related Bessel functions of higher orders; the argument z would of course be set at the largest value likely to be encountered in practice. N is raised successively until the desired accuracy is achieved. This happens when, raising N beyond a certain value, the accuracy is no longer affected. A comparison with the published tables or manual calculations of the Bessel functions of complex variable may be helpful in the beginning. Thus let us suppose that the coefficients (P_n, Q_n) are computed and stored for n = N, N - 1, - - - 3, 2, 1. The higher orders of S_n (z) can now be computed from the definitions of the coefficients themselves as per equation (9) The starting value of S_0 (z) only need be calculated according to equation (2).

The first derivatives of S_n (z) can be evaluated by using the recurrence relation in equation (7). However, the calculations can be considerably economized by using the logarithmic derivatives of S_n (z) and the previous coefficients P_n and Q_n. We define the logarithmic derivative of S_n (z) as :

$$A_n(z) = \frac{S'_n(z)}{S_n(z)} \tag{16}$$

By using equations (7) and (9), this leads to :

$$A_n(z) = (P_n + i\,Q_n)^{-1} - \frac{n}{z} \tag{17}$$

Thus it is easy to compute A_n (z) from equation (17). Then one obtains S_n' (z) with the help of equation (16).

The numerical evaluation of the Riccati-Bessel function of the second kind, C_n (z), and its derivative can be guaranteed to be stable by upward recursion provided the argument is real. However, in the case of complex argument, occasional instability may be encountered. The number of significant digits (*i.e.* precision) in the computer system can impose serious limitations on the use of the upward (forward) recursion. Similar aspect has been discussed by Lentz (1976) in connection with the spherical Bessel function y_n defined by equation (23) in the present article. It may be noted that Lentz's algorithm for the Bessel function of the first kind may break down for certain indexes of refraction as revealed recently by Jaaskelainen and Ruuskanen (1981). Therefore, for more accurate and reliable results, we have adopted the use of Wronskain relation :

$$S_{n-1}(z)\,C_n(z) - S_n(z)\,C_{n-1}(z) = 1.0 \tag{18}$$

Dividing equation (18) throughout by $S_{n-1}(z)$ and using equation (9), one obtains :

$$C_n(z) = (P_n + i\,Q_n)\,C_{n-1}(z) + \frac{1}{S_{n-1}(z)} \tag{19}$$

The coefficients P_n and Q_n have already been evaluated and stored while computing $S_n(z)$ for n = 0 (1) N. We begin by calculating $C_0(z)$ according to equation (5). Then we can initiate equation (19) to compute $C_n(z)$ successively for n = 1 (1) N. Finally the first derivatives of $C_n(z)$ can be evaluated according to equation (8).

Next, the Bessel functions $J_{n+\frac{1}{2}}(z)$ and $Y_{n+\frac{1}{2}}(z)$ as defined by Wats n (1966) and McLachlan (1961) can now be calculated according to their definitions in terms of the above Riccati-Bessel functions as follows :

$$J_{n+\frac{1}{2}}(z) = \left(\frac{2}{\pi z}\right)^{\frac{1}{2}} S_n(z) \tag{20}$$

$$Y_{n+\frac{1}{2}}(z) = -\left(\frac{2}{\pi z}\right)^{\frac{1}{2}} C_n(z) \tag{21}$$

with n = 0, 1, . . . N.

Furthermore, the Bessel functions $j_n(z)$ and $y_n(z)$ defined by Abramowitz and Stegun (1964) can be evaluated from the following relations :

$$j_n(z) = \left(\frac{\pi}{2z}\right)^{\frac{1}{2}} J_{n+\frac{1}{2}}(z) \tag{22}$$

and

$$y_n(z) = \left(\frac{\pi}{2z}\right)^{\frac{1}{2}} Y_{n+\frac{1}{2}}(z) \tag{23}$$

Sample Results and Discussion

The Fortran Program BELJAY, given in Listing 1, is based on the present scheme of calculating the Bessel and related functions. Important FORTRAN words have been explained in Table 1 (page 115). The phase θ of the complex argument z must be set in the correct quadrant because the computer may solve only a right angled triangle in evaluating the library subroutine function θ_e = ATAN (x_2/x_1). The computed value of θ_e may require some adjustment depending on the individual signs of x_1 and x_2. We have derived manually sample results for $j_n(z)$, $y_n(z)$, $J_{n+\frac{1}{2}}(z)$ $Y_{n+\frac{1}{2}}(z)$ with n = 0, 1 and 2 by using their definitions in terms of elementary trignometric functions. These independent results were very useful for checking purposes in the beginning. For use with the modern microcomputers, the program BELJAY may be be modified suitably. For example, the dimensions of quantities such as P_n, Q_n, S_n, C_n etc can be reduced depending on the storage capacity of the available computer. This also means a restriction on the highest permissible argument. The two sets of statements with serial numbers from 112 to 118 and from 178 to 195 provide built-in checks of the accuracy of the functions [$S_n(z)$, $C_n(z)$] and [$j_n(z)$, $y_n(z)$] , respectively. These checks, based on Wronskian relations, work out correctly only if the precision of the computer is sufficiently large. Otherwise loss of leading significant digits in subtraction of nearly equal quantities may result in garbage even if the calculated functions themselves are correct. Therefore, indiscrete use and interpretation of such formulae must be avoided. Some numerical results on the spherical Bessel functions have been summarized in Table 2 to 5 for displaying the general trends and for checking purposes. In each column, an entry consists of 7 or 8 digit decimal number followed by a decimal mantissa. A further check on the correctness of the numerical values of the spherical Bessel functions in various quadrants can be made by using the formulae based on analytical continuation (see, for example, Abramowitz and Stegun 1964 ; McLachlan 1961). These formulae can be expressed in a more direct manner as follows :

$$\mathrm{Re}\, J_{n+\frac{1}{2}}(-z) = (-1)^{n+1}\, \mathrm{Im}\, J_{n+\frac{1}{2}}(+z) \tag{24}$$

$$\mathrm{Im}\, J_{n+\frac{1}{2}}(-z) = (-1)^{n}\, \mathrm{Re}\, J_{n+\frac{1}{2}}(+z) \tag{25}$$

$$\mathrm{Re}\, Y_{n+\frac{1}{2}}(-z) = (-1)^{n}\, \mathrm{Im}\, Y_{n+\frac{1}{2}}(+z) \tag{26}$$

and

$$\mathrm{Im}\, Y_{n+\frac{1}{2}}(-z) = (-1)^{n+1}\, \mathrm{Re}\, Y_{n+\frac{1}{2}}(+z) \tag{27}$$

Table 2. Spherical Bessel Functions of orders $n+\frac{1}{2}$ with argument $z = 5.0 + i\,2.0$

n	Re $J_{n+\frac{1}{2}}$	Im $J_{n+\frac{1}{2}}$	Re $Y_{n+\frac{1}{2}}$	Im $Y_{n+\frac{1}{2}}$
0	−1.1511382+000	5.8191959−001	−5.8644269−001	−1.1048265+000
1	−7.4478241−001	−9.2510678−001	9.7383243−001	−7.3196260−001
2	5.7450450−001	−9.0633016−001	9.3870858−001	5.2474262−001
3	9.2751726−001	−5.4317374−002	1.6344841−002	8.6063432−001
4	5.1869071−001	3.9300776−001	−5.0350342−001	5.0606336−001
5	1 2128348−001	3.4221105−001	−4.8353495−001	2.3715577−001
10	−1.6100112−003	−6.8509214−004	1.8231363+001	−5.4395387+000
20	1.3248081−011	4.4603493−011	−1.0562978+008	3.2546731+008
50	2.1261187−044	5.8553692−045	−2.7700009+041	7.5108888+040
100	1.5238673−116	8.1769928−117	−1.6163144+113	8.6524146+112

Similar relations exist in the case of $j_n(+z)$ and $y_n(+z)$. The results for a variety of complex variables $z = \pm x_1 \pm i x_2$ lying in all the four quadrants conform to the above formulae (24-27). Furthermore, it has been found that the results for any one quadrant are sufficient to evaluate the Bessel functions in any other quadrant provided the argument in either case has the same numerical values for corresponding real as well as imaginary parts. Therefore, we have given the results in Table 2 to 5 only for the first or fourth quadrant.

Table 3. Spherical Bessel Functions of orders $n+\frac{1}{2}$ with argument $z = 10.0 - i\,10.0$

n	Re $J_{n+\frac{1}{2}}$	Im $J_{n+\frac{1}{2}}$	Re $Y_{n+\frac{1}{2}}$	Im $Y_{n+\frac{1}{2}}$
0	−1.924736E+03	1.324923E+03	1.324923E+03	1.924736E+03
1	1.162440E+03	1.894745E+03	1.894745E+03	−1.162440E+03
2	1.814890E+03	−8.663449E+02	−8.663449E+02	−1.814890E+03
3	−4.921310E+02	−1.657609E+03	−1.657609E+03	4.921310E+02
4	−1.406973E+03	1.139360E+02	1.139360E+02	1.406973E+03
5	−1.922778E+02	1.075742E+03	1.075742E+03	1.922778E+02
10	−7.048671E+01	1.099629E+02	1.099630E+02	7.048678E+01
20	8.956464E−03	−2.478288E−02	−3.263045E−01	−4.604763E−01
50	2.016208E−23	−3.091142E−23	−9.868115E+19	−1.390764E+20
99	−2.257627E−73	−2.788426E−73	5.639975E+69	−6.955606E+69
100	−2.511649E−74	−2.517626E−75	1.247062E+71	−1.374863E+70

Table 4. Spherical Bessel Functions of orders $n+\frac{1}{2}$ and argument $z=100.0-i\,10.0$

n	Re $J_{n+\frac{1}{2}}$	Im $J_{n+\frac{1}{2}}$	Re $Y_{n+\frac{1}{2}}$	Im $Y_{n+\frac{1}{2}}$
0	−4.056493E+02	−7.770334E+02	−7.770334E+02	4.056493E+02
1	−7.802804E+02	3.975543E+02	3.975543E+02	7.802804E+02
2	3.812918E+02	7.865243E+02	7.865243E+02	−3.812918E+02
3	7.952626E+02	−3.567298E+02	−3.567298E+02	−7.952626E+02
4	−3.237022E+02	−8.057364E+02	−8.057364E+02	3.237022E+02
5	−8.169275E+02	2.820471E+02	2.820471E+02	8.169275E+02
10	−5.357223E+01	8.304771E+02	8.304771E+02	5.357223E+01
20	7.174170E+02	2.606982E+01	2.606982E+01	−7.174170E+02
50	3.232162E+01	2.411763E+02	2.411764E+02	−3.232161E+01
100	−3.779284E−01	−2.727924E−01	−2.695123E−01	3.630760E−01
150	1.122696E−16	4.403574E−16	−1.054084E+12	6.075059E+12

Table 5. Spherical Bessel Functions of orders $n+\frac{1}{2}$ with argument $z=100.0-i\,100.0$

n	Re $J_{n+\frac{1}{2}}$	Im $Y_{n+\frac{1}{2}}$	Re $Y_{n+\frac{1}{2}}$	Im $Y_{n+\frac{1}{2}}$
0	−1.242885E+41	−8.931742E+41	−8.931742E+41	1.242885E+41
1	−8.893298E+41	1.192012E+41	1.192012E+41	8.893298E+41
2	1.091605E+41	8.816223E+41	8.816223E+41	−1.091605E+41
3	8.700182E+41	−9.443163E+40	−9.443163E+40	−8.700182E+41
4	−7.540480E+40	−8.544767E+41	−8.544767E+41	7.540480E+40
5	−8.349600E+41	5.258696E+40	5.258696E+40	8.349600E+41
10	−9.420728E+40	6.783730E+41	6.783730E+41	9.420728E+40
20	2.496741E+41	−1.920328E+41	−1.920328E+41	−2.496741E+41
50	1.216043E+38	1.431730E+39	1.431730E+39	−1.216043E+38
100	−3.486461E+30	−1.843787E+30	−1.843787E+30	3.486461E+30
150	−2.570873E+15	7.674092E+15	7.674092E+30	2.570873E+15
200	1.510023E−05	−2.088504E−06	−4.405181E+01	−3.818836E+01
220	5.831385E−15	2.418326E−16	−2.312352E+11	5.574836E+10

If z lies on the negative x-axis such that $-z = -x_1 + i\,0.0$, the equations (24−27) are still valid. However, for z on the positive or negative y-axis, we may define $z_3 = 0.0 + i\,|x_3|$ and $z_4 = 0.0 - i\,|x_3|$. It is found that the following equalities hold :

$$\mathrm{Re}\, J_{n+\frac{1}{2}}(z_3) = (-1)^n\, \mathrm{Im}\, J_{n+\frac{1}{2}}(z_3) \tag{28}$$

$$\mathrm{Re}\, Y_{n+\frac{1}{2}}(z_3) = (-1)^{n+1}\, \mathrm{Im}\, Y_{n+\frac{1}{2}}(z_3) \tag{29}$$

$$\overline{J_{n+\frac{1}{2}}(z_3)} = J_{n+\frac{1}{2}}(z_4) \tag{30}$$

$$\overline{Y_{n+\frac{1}{2}}(z_3)} = Y_{n+\frac{1}{2}}(z_4) \tag{31}$$

where a bar over a function indicates the complex conjugate.

From a detailed examination of the Tables 2 to 5 and related results for other quadrants, we have noticed a remarkable result which is not quite obvious analytically. Provided that $|\text{Im}(z)|$ is large compared to unity and $|\text{Re}(z)|$ is not too near zero, the following approximate relations are satisfied upto a certain value of $n = n_{max}$. For quadrants 1 and 2,

$$\text{Re}\, J_{n+\frac{1}{2}}(z) \simeq \text{Im}\, Y_{n+\frac{1}{2}}(z) \tag{32}$$

and

$$\text{Im}\, J_{n+\frac{1}{2}}(z) \simeq -\text{Re}\, Y_{n+\frac{1}{2}}(z) \tag{33}$$

For quadrants 3 and 4,

$$\text{Re}\, J_{n+\frac{1}{2}}(z) \simeq -\text{Im}\, Y_{n+\frac{1}{2}}(z) \tag{34}$$

and

$$\text{Im}\, J_{n+\frac{1}{2}}(z) \simeq \text{Re}\, Y_{n+\frac{1}{2}}(z) \tag{35}$$

The relations in equation (32-35) have been verified analytically also. Therefore, they are useful for checking purpose. The highest order n_{max} upto which equations (32-35) are valid is set out in Table 6 for various argument $z = \pm x_1 \pm ix_2$. Note that we have considered at least the first two significant digits to be identical. This constraint implies that we have imposed equality signs in equations (32-35) at least for the first two significant digits.

Table 6. The highest order upto which equations (32–35) hold.

$\lvert \text{Re } z \rvert$	$\lvert \text{Im } z \rvert$	n_{max}
0.01	2.0	0
1.0	5.0	4
2.0	5.0	4
5.0	2.0	0
10.0	5.0	10
10.0	10.0	16
100.0	5.0	80
100.0	10.0	96
100.0	100.0	190
500.0	40.0	516

Finally we show a sample comparison with the work of Lentz (1981). We choose a common set of values for complex argument z and the orders n, viz.

$$z = 10 - i\,10,$$

and

$$n = 40, 41.$$

The cross product formula reads

$$j_n(z)\, y_{n-1}(z) - j_{n-1}(z)\, y_n(z) = \frac{1}{z^2} \tag{36}$$

The real and imaginary parts of $\frac{1}{z^2}$ are 0.0 and 5.0E −03, respectively, whereas the results of calculating the L.H.S of equation (36) are shown below :

Author	L.H.S. of equation (36) Real Part	Imaginary part
Lentz	2.3857835E — 11	−4.99 — — — ?
Shah (present work)	−6.9388939E — 18	−4.9999998E — 3

It is now clear that the present set up satisfies the cross product test of equation (36) far better as compared to the work of Lentz.

Practical Aspects

While working with an actual physical problem, one may need Bessel functions and their derivatives of orders n from zero upto a certain maximum, say $n = N_o$. N_o depends on the complex argument z among other things. As an illustration, let us consider the Mie theory of scattering of electromagnetic wave by a smooth, homogeneous and isotropic sphere of radius a and composed of material having refractive index $m = m' - im''$ (see, for example, van de Hulst 1957). The relevant argument in this case can be defined by $z = m\alpha$, where $\alpha = 2\pi\, a/\lambda$, λ being the wavelength of the incident electromagnetic radiation. If $z << 1$, the Rayleigh scattering domain prevails and usually the first few terms *i.e.* $N_o = 2$ to 5, are adequate for evaluating the Mie coefficients accurate to 10 significant digits. All the radiation scattering parameters can be derived from the Mie coefficients. However, for $z \gtrsim 1$, one is concerned with the anomalous scattering and geometrical diffraction region. Here the value of N_o depends on α and m both (see, for example, Shah 1977). Let us consider a set of size-to-wavelength parameter, $\alpha = 1$, 10, 100 and 1000. For dirty ice spheres with $m = 1.33 - i\, 0.05$ in the visual wavelength region, one has $N_0 = 8$, 21, 124, and 1047, respectively. For ice with $m = 1.78 - i\ 0.0024$ in millimeter and centimeter microwave regions, corresponding $N_0 = 7$, 20, 123 and 1046. For rain drops at $\lambda = 3$ millimeters, $m = 3.41 - i\ 1.94$ and for the same set of α, N_0 turns out to be 7, 22, 123 and 1047, respectively. Thus the highest order of Bessel functions needed in Mie theory calculations is dictated mainly by the ratio of the size of the scattering particle to the wavelength of the incident radiation. An empirical relation for $N_0 = N_0\,(\alpha, m', m'')$ has been given explicitly by Shah (1977).

Conclusion

The numerical evaluation of the spherical Bessel and related functions, even beyond the range of validity of the conventional asymptotic expansions, can be effected efficiently by the present techniques. It is hoped that this work will find wide application in variety of fields such as Physics, Astronomy and Astrophysics.

Acknowledgements

It is a pleasure to thank Dr. K. S. Krishna Swamy for assistance, Dr. S. S. Hasan for his interest in this work, and Dr. J. C. Bhattacharyya for reading the manuscript. I am grateful to Dr. Irene A. Stegun for providing independent calculations for $z = \pm 5.0 \pm i\, 2.0$ and $n = 0\ (1)\ 21$.

References

Abramowitz, M., Stegun, I.A., 1964, *Handbook of Mathematical Functions with Formulas, Graphs and Mathematical Tables*, National Bureau of Standards, Washington.
Brouwer, D., Clemence, G.M., 1961, Method of Celestial Mechanics, Academic Press, New York, p. 81.
Gautschi, W., 1967, *SIAM Review*, 9, 24.
Jaaskelainen, T., Ruuskanen, J., 1981, *Appl. Optics*, 20, 3289.
Lentz, W.J., 1976, *Appl. Optics*, 15, 668.
Lentz, W.J., 1981, Personal Communication from Journal of Computational Physics.
McLachlan, N.W., 1961, *Bessel Functions for Engineers*, Oxford University Press.
Shah, G. A., 1967, Ph.D. Thesis : *Models of Interstellar Grains*, Rensselaer Polytechnic Institute, Troy, New York.
Shah, G. A., 1970, *Mon. Not. R. astr. Soc.*, 148, 93.
Shah, G. A., 1977, *Kodaikanal Obs. Bull Ser. A.*, 2, 42.
Temme, N. M., 1976, *J. Comp. Phys.*, 21 343.
Van de Hulst, H.C., 1957, *Light Scattering by Small Particles*, John Wiley & Sons, New York.
Watson, G.N., 1966, *A Treatise on the Bessel Functions*, Cambridge University Press.

Table 1 : Explanation of some Fortran words in the computer program 'BELJAY'.

Fortran word	Equation No.	Meaning or equivalence in the text.
NMAX	(9), (12-15) (20-23)	The arguments of the dimensioned variables P_n, Q_n, Re $J_{n+\frac{1}{2}}(z)$, Im $J_{n+\frac{1}{2}}(z)$, etc.
Y_1, Y_2		The real (x_1) and imaginary (x_2) parts of the complex argument z. These are entered through READ statement number 1.
NN		Maximum order upto which Bessel functions are to be evaluated.
N	(12 to 15)	The highest value of n adopted for starting the downward recursion relations.
NX		Order upto which P_n, Q_n are stored; NX depends on the absolute value of z. All functions are calculated upto this order, or upto NN.
NS, CN		n, current order of Riccati-Bessel functions. Note that the zero ordered functions are evaluated above the loops "DO 10", "DO 11" and "DO 12".
SR (NS) SI (NS) CR (NS) CI (NS)	(9) (19)	The real and imaginary parts of Riccati-Bessel functions.
DSR (NS) DSI (NS) DCR (NS) DCI (NR)	(16) (8)	The same for the first derivatives.
TH		The phase of the argument z, located in the appropriate quadrant in the z-plane.
BJZR BJZI		The real and imaginary parts of $J_{\frac{1}{2}}(z)$.
BYZR BYZI		The same for $Y_{\frac{1}{2}}(z)$.
BJR (NS) BJI (NS)	(20)	The real and imaginary parts of $J_{n+\frac{1}{2}}(z)$, n = NS.
BYR (NS) BYI (NS)	(21)	The same for $Y_{n+\frac{1}{2}}(z)$.
ASJZR ASJZI		The real and imaginary parts of $j_0(z)$.
ASYZR ASYZI		The same for $y_0(z)$.
ASJR (NS) ASJI (NS)	(22)	The real and imaginary parts of $j_n(z)$, n = NS
ASYR (NS) ASYI (NS)	(23)	The same for $y_n(z)$.

LISTING 1

Fortran Program BELJAY for Calculating

Riccati-Bessel and Bessel Functions

```
      PROGRAM BELJAY(OUTP,OUTPUT=OUTP)
      DIMENSION P(2001),Q(2001),SR(2001),SI(2001)
      DIMENSION DSR(2001),DSI(2001),BJR(2001),BJI(2001)
      DIMENSION BYR(2001),BYI(2001)
      DIMENSION CR(2001),CI(2001),DCR(2001),DCI(2001)
      DIMENSION ASJR(2001),ASJI(2001),ASYR(2001),ASYI(2001)
  500 FORMAT(1X,'INDETERMINATE FORM T=ATAN(0/0) AT STATEMENT  NO.73')
  502 FORMAT(/1X,'--------------------------------------------------------
     1---------------------------------')
  800 FORMAT(1X,I5,4(2X,E15.7))
  801 FORMAT(8F10.5)
  802 FORMAT(//10(5X,I5)//)
  803 FORMAT(5X,'NEXT LOT GIVES BESSEL FUNCTIONS AS TABULAIED BY ABRAMO
     1WITZ AND STEGUN(DOVER 1965),PAGES 457-466')
  804 FORMAT(9X,'RE(S)',7X,'IM(S',7X,'RE(C)',7X,'IM(C)',4X,'NS',
     15X,'AND THEIR RESPECTIVE DERIVATIVES')
  805 FORMAT(5X,'N',5X,'REJ(N+1/2)',6X,'IMJ(N+1/2)',6X,'RE Y(N+1/2)',
     16X,'IM Y(N+1/2)')
  806 FORMAT(2X,2(5X,I5),5X,4(2X,E15.8))
  810 FORMAT(//10X,'ARGUMENT AND/OR ORDER TOO LARGE,DIMENSIONS NEED LAR
     1GER NUMBERS,VIZ.(NN+1).CALCULATIONS NOT ATTEMPTED.'//)
  881 FORMAT(/5X,'NMAX',7X,'NN',7X,'NX',7X,'N'/)
  888 FORMAT(/5X,'REAL OF Z=',E11.4,10X,'IMAGINARY OF Z=',E11.4)
      NMAX=2001
       OPEN(UNIT=2,FILE='BELDATA',STATUS='OLD')
    1 READ(2,801)Y1,Y2
      IF(10000.-Y1)999,999,1000
 1000 CONTINUE
C     IN SUBROUTINE FORM THIS PROGRAM MAY BE CALLED SUBROUTINE BELJAY(X,
C     Y,NN). REMOVE THREE STATEMENTS 1 TO 1000. INSTEAD SET Y1=X, Y2=Y.
C     IN THE BEGINNING OF THE PROGRAM.ALSO REMOVE STATEMENT NN=100.
      NN=100
      YY=Y1*Y1+Y2*Y2
      Y=SQRT(YY)
      NYY=1.5*Y+10.0
      NX=NYY
      PRINT 800
      PRINT 800
      IF(NN-NX) 22,22,21
   21 NX=NN
   22 IF(NX-NMAX)29,29,28
   28 PRINT 810
      GO TO 999
   29 CONTINUE
```

```
   31 N=NX+0.5*Y+50.0
   32 PJN1=0.0
      QJN1=0.0
      JN=N+1
   35 JN=JN-1
      XN=2*JN+1
      YN=XN/YY
      PR=YN*Y1-PJN1
      PI=YN*Y2+QJN1
      PP=PR*PR+PI*PI
      PJN=PR/PP
      QJN=PI/PP
      IF(JN-NX)38,38,39
   38 P(JN)=PJN
      Q(JN)=QJN
   39 PJN1=PJN
      QJN1=QJN
      IF(JN-1)40,40,35
   40 CONTINUE
C     COEFFICIENTS P(N) AND Q(N) IN DOWNWARD RECURSION ON RICCATI-BESSEL
C     FUNCTION OF FIRST KIND ARE NOW READY AND STORED UP TO ORDERS N=NX.
C     NOW CALCULATE RICCATI-BESSEL FUNCTIONS AND THEIR DERIVATIVES.
      SNY=SIN(Y1)
      CSY=COS(Y1)
      XEP=EXP(Y2)
      XEQ=EXP(-Y2)
      CHY=0.5*(XEP+XEQ)
      SHY=0.5*(XEP-XEQ)
      CCHY=CHY*CSY
      SSHY=SHY*SNY
      SCHY=SNY*CHY
      CSHY=CSY*SHY
      CNZR=CCHY
      CNZI=-SSHY
      SZR=SCHY
      SZI=CSHY
      PRINT 881
      PRINT 802,NMAX,NN,NX,N
      PRINT 802
      DO 10 NS=1,NX
      CN=NS
      CNYR=-CN*Y1/YY
      CNYI=CN*Y2/YY
      IF(NS-1)7,7,8
    7 SR(1)=P(1)*SZR-Q(1)*SZI
      SI(1)=P(1)*SZI+Q(1)*SZR
      CDEN=SZR**2+SZI**2
      CR(1)=P(1)*CNZR-Q(1)*CNZI+SZR/CDEN
      CI(1)=P(1)*CNZI+Q(1)*CNZR-SZI/CDEN
      DSR(1)=SZR-(SR(1)*Y1+SI(1)*Y2)/YY
      DSI(1)=SZI+(SR(1)*Y2-SI(1)*Y1)/YY
      DCR(1)=CR(1)*CNYR-CI(1)*CNYI+CNZR
      DCI(1)=CR(1)*CNYI+CI(1)*CNYR+CNZI
```

```
      GO TO 10
    8 SR(NS)=P(NS)*SR(NS-1)-Q(NS)*SI(NS-1)
      SI(NS)=P(NS)*SI(NS-1)+Q(NS)*SR(NS-1)
      SRSI=SR(NS-1)**2+SI(NS-1)**2
      CR(NS)=P(NS)*CR(NS-1)-Q(NS)*CI(NS-1)+SR(NS-1)/SRSI
      CI(NS)=P(NS)*CI(NS-1)+Q(NS)*CR(NS-1)-SI(NS-1)/SRSI
      PQ=(P(NS)**2)+(Q(NS)**2)
      PSR=P(NS)/PQ-CN*(Y1/YY)
      PSI=CN*(Y2/YY)-Q(NS)/PQ
      DSR(NS)=SR(NS)*PSR-SI(NS)*PSI
      DSI(NS)=SR(NS)*PSI+SI(NS)*PSR
      DCR(NS)=CR(NS)*CNYR-CI(NS)*CNYI+CR(NS-1)
      DCI(NS)=CR(NS)*CNYI+CI(NS)*CNYR+CI(NS-1)
   10 CONTINUE
      MX=NX-1
      DO 50 I=1,MX
      K=I+1
      TSCR=SR(I)*CR(K)-SI(I)*CI(K)-SR(K)*CR(I)+SI(K)*CI(I)
      TSCI=SR(I)*CI(K)+SI(I)*CR(K)-SR(K)*CI(I)-SI(K)*CR(I)
      PRINT 806,I,K,TSCR,TSCI
50    CONTINUE
C     NEXT COMPUTE THE USUAL BESSEL J(N+1/2) AND Y(N+1/2)FUNCTIONS
      THETA=ATAN(Y2/Y1)
      IF(Y1)75,74,77
   74 IF(Y2)71,73,72
   71 THETA=-1.5707963
      GO TO 77
   72 THETA=1.5707963
      GO TO 77
   73 PRINT 500
      GO TO 999


   75 IF(Y2)76,78,76
   76 THETA=3.14159265+THETA
      GO TO 77
   78 THETA=3.14159265
   77 TH=THETA/2.0
      TPR=0.79788456080286536/SQRT(Y)
      AR=TPR*COS(TH)
      AI=-TPR*SIN(TH)
      NS=0
      BJZR=AR*SZR-AI*SZI
      BJZI=AR*SZI+AI*SZR
      BYZR=-(AR*CCHY+AI*SSHY)
      BYZI=AR*SSHY-AI*CCHY
      PRINT 802
      PRINT 888,Y1,Y2
      PRINT 502
      PRINT 805
      PRINT 502
      PRINT 800,NS,BJZR,BJZI,BYZR,BYZI
      DO 11 NS=1,NX
      BJR(NS)=AR*SR(NS)-AI*SI(NS)
```

```
      BJI(NS)=AR*SI(NS)+AI*SR(NS)
      BYR(NS)=-(AR*CR(NS)-AI*CI(NS))
      BYI(NS)=-(AR*CI(NS)+AI*CR(NS))
      PRINT 800,NS,BJR(NS),BJI(NS),BYR(NS),BYI(NS)
   11 CONTINUE
      NS=0
      PBT=1.5707963
      ASJZR=PBT*(AR*BJZR-AI*BJZI)
      ASJZI=PBT*(AR*BJZI+AI*BJZR)
      ASYZR=PBT*(AR*BYZR-AI*BYZI)
      ASYZI=PBT*(AR*BYZI+AI*BYZR)
      PRINT 802
      PRINT 802
      PRINT 502
      PRINT 803
      PRINT 502
      PRINT 802
      PRINT 888,Y1,Y2
      PRINT 802
      PRINT 802
      PRINT 800,NS,ASJZR,ASJZI,ASYZR,ASYZI
      DO 12 NS=1,NX
      ASJR(NS)=PBT*(AR*BJR(NS)-AI*BJI(NS))
      ASJI(NS)=PBT*(AR*BJI(NS)+AI*BJR(NS))
      ASYR(NS)=PBT*(AR*BYR(NS)-AI*BYI(NS))
      ASYI(NS)=PBT*(AR*BYI(NS)+AI*BYR(NS))
      PRINT 800, NS,ASJR(NS),ASJI(NS),ASYR(NS),ASYI(NS)
   12 CONTINUE
      PRINT 502
      AZZ=YY*YY
      ZZR=(Y1**2-Y2**2)/AZZ
      ZZI=-(2.0*Y1*Y2)/AZZ
      TJYR=ASJR(1)*ASYZR-ASJI(1)*ASYZI-ASJZR*ASYR(1)+ASJZI*ASYI(1)
      TJYI=ASJR(1)*ASYZI+ASJI(1)*ASYZR-ASJZR*ASYI(1)-ASJZI*ASYR(1)
      I=0
      N2=1
      PRINT 806,I,N2,TJYR,TJYI,ZZR,ZZI
      MX=NX-1
      DO 555 I=1,MX
      K=I+1
      TJYR=ASJR(K)*ASYR(I)-ASJI(K)*ASYI(I)-ASJR(I)*ASYR(K)+ASJI(I)*
     1ASYI(K)
      TJYI=ASJR(K)*ASYI(I)+ASJI(K)*ASYR(I)-ASJR(I)*ASYI(K)-ASJI(I)*
     1ASYR(K)
      PRINT 806,I,K,TJYR,TJYI
  555 CONTINUE
      GO TO 1
  999 CONTINUE
      STOP
      END
```